Lundi 1er novembre 2004 à 17 h

Volume d'un corps de révolution

Bernard Vuilleumier

Calculer le volume V d'un corps de révolution engendré par la rotation d'un profil autour d'un axe est une des tâches du calcul intégral. Si le profil est donné par une fonction $y = f(x)$ avec x variant entre a et b, le volume du corps se calcule à l'aide d'une intégrale définie. Cette intégrale définie correspond à la somme des volumes élémentaires $v_i = \pi y_i^2\, \Delta x$ de n « rondelles » de rayon y_i constant et d'épaisseur Δx, lorsque le nombre n de rondelles devient très grand et leur épaisseur Δx très petite :

$$V \simeq \sum_{i=1}^{n} \pi y_i^2\, \Delta x \qquad\qquad V = \int_a^b \pi y^2\, dx$$



Fig. 1 : Approximation du volume d'un corps de révolution par le volume des « rondelles » qui le composent. Le volume du corps s'obtient par une intégrale définie qui fait tendre le nombre de rondelles vers l'infini et leur épaisseur vers zéro.

Mais il existe de nombreuses situations, aussi bien en sciences expérimentales qu'en mathématiques numériques, dans lesquelles on ne dispose pas de l'expression symbolique $y = f(x)$ de la fonction à intégrer, mais seulement d'une liste y_i de valeurs. La commande **`ListIntegrate`** du module **`` NumericalMath`ListIntegrate` ``** calcule dans ce cas une approximation de l'intégrale. **`ListIntegrate`** travaille en utilisant une interpolation pour construire un « objet » qui approche la fonction à intégrer à l'aide de polynômes. Vous pouvez spécifier le degré des polynômes de cet objet en donnant le nombre de points utilisés pour les construire. Si k est le nombre de points utilisés pour construire un polynôme, le degré du polynôme est k - 1. La valeur par défaut pour k est 4. L'approximation peut être faite à partir d'une liste de valeurs $\{y_1, y_2, \ldots\ y_n\}$ régulièrement espacées selon Ox, ou à partir d'une liste de coordonnées $\{\{x_1, y_1\}, \ldots\ \{x_n, y_n\}\}$ pour des valeurs non régulièrement espacées selon Ox.



Fig. 2 : Le volume d'un corps de révolution peut être obtenu à partir de valeurs y_i régulièrement espacées ou non selon Ox.

Prochaine réunion : lundi 6 décembre 2004 à 17 h

Travaux pratiques

■ Exercice 1

Soit le profil défini par $y = \sin x + 1.1$ pour x variant de 1.3 à 6.5 :
a) Dessinez ce profil et une approximation « en escaliers » de ce profil à l'aide de n marches.



b) Calculez le volume exact du corps de révolution engendré par ce profil.
c) Calculez le volume des « rondelles » engendrées par la rotation du profil en escalier autour de Ox.
d) Etablissez le graphique donnant le volume total des rondelles en fonction du nombre de rondelles.

■ Exercice 2

a) Dessinez le profil défini par la fonction $y = x(1 - x)$ lorsque x varie de 0 à 1.
b) Dessinez le corps de révolution engendré par la rotation de ce profil autour de l'axe Ox.
c) Calculez le volume exact de ce corps de révolution.

■ Exercice 3

On donne les cinq points suivants `{{0,0},{0.25,0.75},{0.5,1.},{0.75,0.75},{1.,0.}}`
a) Créez un profil de type « spline cubique » qui passe par ces 5 points.
b) Dessinez le corps de révolution résultant de la rotation de ce profil autour de l'axe Ox.
c) Calculez le volume obtenu à partir de ce profil interpolé et comparez-le au volume obtenu avec $y = x(1 - x)$.
N. B. Les cinq points définissant le profil interpolé appartiennent tous au profil $y = x(1 - x)$ de l'exercice 2.

Volume d'un corps de révolution

■ Exercice 1

a) Dessinons la fonction et son approximation à l'aide de *n* marches :

```
n = 20; (* nombre de marches *)
Needs["Graphics`FilledPlot`"]
Needs["Graphics`Colors`"]
f[x_] := Sin[x] + 1.1; a = 1.3; b = 6.5; Δx = (b - a) / n;
p = Plot[f[x], {x, a, b},
   AspectRatio → Automatic, Ticks → None, AxesOrigin → {a, 0},
   PlotStyle → {Thickness[0.005], Blue}, DisplayFunction → Identity];
pts = Table[{x, f[x]}, {x, a, b, Δx}];
pts2 = Table[{x, f[x - Δx]}, {x, a + Δx, b + Δx, Δx}];
pts3 = Flatten[Transpose[{pts, pts2}], 1];
e = FilledListPlot[Drop[pts3, -2], AspectRatio → Automatic, Ticks → None,
   AxesOrigin → {a, 0}, Fills → Pink, DisplayFunction → Identity];
Show[e, p, DisplayFunction → $DisplayFunction];
```

b) Calculons le volume V du corps de révolution qui est donné par :

$$V = \int_a^b \pi y^2 \, dx$$

```
Integrate[π * f[x] ^ 2, {x, a, b}]
```

c) Calculons l'épaisseur Δx des rondelles, le volume de chaque rondelle et sommons les volumes des rondelles :

```
n = 20; (* nombre de marches *)
Δx = (b - a) / n;
Apply[Plus, Table[π * f[x] ^ 2, {x, a, b, Δx}] * Δx]
```

d) Etablissons le graphique donnant le volume total des rondelles en fonction du nombre *n* de rondelles :

```
ListPlot[
  Table[{n, Apply[Plus, Table[π * f[x] ^ 2, {x, a, b, (b - a) / n}] * (b - a) / n]},
   {n, 100, 1000, 50}], PlotStyle → Hue[1], AxesOrigin → {0, 23.109}];
```

■ Exercice 2

a) Définissons la fonction et dessinons le profil pour x variant de 0 à 1 :

```
Plot[x (1 - x), {x, 0, 1}, PlotRange → All];
```

b) La commande **`SurfaceOfRevolution`** du module de même nom dessine le corps de révolution :

```
Needs["Graphics`SurfaceOfRevolution`"]
SurfaceOfRevolution[x (1 - x), {x, 0, 1},
  Boxed → False, Axes → None, RevolutionAxis → {1, 0, 0}];
```

c) Calculons le volume exact V engendré par la révolution de la courbe $y = f(x)$ pour x variant de 0 à 1 :

$$V = \int_0^1 \pi y^2 \, dx$$

```
Integrate[π * (x (1 - x)) ^ 2, {x, 0, 1}]
N[%]
```

$\frac{\pi}{30}$

▪ Exercice 3

a) Affectons les points à la variable **pts**, chargeons le module **Graphics`Spline`** et dessinons le profil :

```
pts = Table[{x, x (1 - x)}, {x, 0, 1, 0.25}];
Needs["Graphics`Spline`"]
Show[Graphics[Spline[pts, Cubic, SplineDots → Automatic]], Axes → True];
```



b) Ici, le profil n'est plus donné sous la forme $y = f(x)$. Pour dessiner le corps de révolution, nous devons au préalable établir une liste de points à partir de la fonction paramétrique fournie par **SplineFit** en faisant varier le paramètre de 0 à 4. Nous utilisons ensuite **ListSurfaceOfRevolution** :

```
Needs["NumericalMath`SplineFit`"]
s = SplineFit[pts, Cubic]
```

```
SplineFunction[Cubic, {0., 4.}, <>]
```

```
ListSurfaceOfRevolution[Table[s[t], {t, 0, 4, 0.1}],
  Boxed → False, Axes → None, RevolutionAxis → {1, 0, 0}];
```



c) Pour calculer le volume d'un corps de révolution défini par un profil donné à l'aide d'une liste de points, nous utilisons **ListIntegrate** du module **NumericalMath`ListIntegrate`**. Nous définissons une fonction g qui associe à chaque point (x, y) du profil le point $(x, \pi y^2)$. En appliquant cette fonction aux points du profil puis en intégrant la liste, nous obtenons un volume qui ne diffère que de 0.53 % de la valeur exacte :

```
Needs["NumericalMath`ListIntegrate`"]
g[{x_, y_}] := {x, π * y^2}
ListIntegrate[Map[g, Table[s[t], {t, 0, 4, 0.1}]]]
Abs[Integrate[π * (x (1 - x))^2, {x, 0, 1}] - %] /
 Integrate[π * (x (1 - x))^2, {x, 0, 1}]
```

```
0.104164
```

```
0.00530862
```